
*=======PHASE_MODULATE_USING_JITTER=============

A free running oscillator can export its jitter into 
a piece wise linear noise file. This can be used to 
phase modulate a square wave in order to compare
spectrums.

Use_Jitter_to_PM
*    OUT     Rload 
*     _____/\  __          
*   _|_      \/  |         
*  /   \         |      
* /VpwlT\        |          
* \     /       _|_     
*  \___/        ///     
*    |          Gnd 
*   _|_                 
*   /// 
*   Gnd     
*
* VpwlT OUT 0        PWL( + 0.0005 0.988835 +.....
*=========Include_Noise_File=================================================
.include  PWL_FileJitter.inc
Rload     OUT    0      1k
VT        Vtime  0      PWL    ( 0 0 2 2 ) DC 0
B1        OUT1   0      V =     cos(2*3.14159*20*v(Vtime)+ v(OUT))
B2        OUT2   0      V =     20*u(v(OUT1))-10
*=======================================================================

The Noise is being applied in a radian magnitude format.

*=========Run_Simulation=================================================
*TRAN     TSTEP  TSTOP  TSTART TMAX      ?UIC?
.tran     5u     2      0      5u        UIC
.control
run
set       pensize =     2
plot      out1  out
*=======================================================================

The Noise is applied to a unity gain cosine wave 
which is then clipped.

*===========Find_Spectrum===================================
linearize 
set         specwindow =    "rectangular" 
*SPEC       FSTART  FSTOP    FSTEP   VECTOR
spec        .5      1000    .5       v(OUT2) 
plot        mag(v(OUT2))    loglog
*============================================================

The following is the resulting spectrum when phase
modulating a 20Hz with a jitter PWL file.



Below is the spectrum of the free running oscillator
which generated the jitter PWL file. 

Two very different paths were taken to get to the same 
spectrum. So things are very close, but not identical. 

For a more detailed view of the random noise sidebands,
it is possible to estimate where these noise sidebands
come from.

===========Full_Netlist_For_Copy_Paste=======================
Use_Jitter_to_PM
*    OUT     Rload 
*     _____/\  __          
*   _|_      \/  |         
*  /   \         |      
* /VpwlT\        |          
* \     /       _|_     
*  \___/        ///     
*    |          Gnd 
*   _|_                 
*   /// 
*   Gnd     
*
* VpwlT OUT 0        PWL( + 0.0005 0.988835 +.....
*=========Include_Noise_File=================================================
.include  PWL_FileJitter.inc
Rload     OUT    0      1k
VT        Vtime  0      PWL    ( 0 0 2 2 ) DC 0
B1        OUT1   0      V =     cos(2*3.14159*20*v(Vtime)+ v(OUT))
B2        OUT2   0      V =     20*u(v(OUT1))-10
*=========Run_Simulation=================================================



*TRAN     TSTEP  TSTOP  TSTART TMAX      ?UIC?
.tran     5u     2      0      5u        UIC
.control
run
set       pensize =     2
plot      out1  out
*===========Find_Spectrum===================================
linearize 
set         specwindow =    "none" 
*SPEC       FSTART  FSTOP    FSTEP   VECTOR
spec        .5      1000    .5       v(OUT2) 
plot        mag(v(OUT2))    loglog
*=========Wrap_Up=======================================================
.endc
.end

